Tuned Mass Damper Design for Attenuating Vibration

Introduction

A mass-spring-damper is disturbed by a force that resonates at the natural frequency of the system. This application calculates the optimum spring and damping constant of a parasitic tuned-mass damper that the minimizes the vibration o the system.

The vibration of system with and without the tuned mass-spring-damper is viewed as a frequency response, time-domain simulation and power spectrum.



> restart : with(DynamicSystems) : with(plots) : with(SignalProcessing) :

> params := $\left[m_1 = 1.764\,10^5, k_1 = 3.45\,10^7, b_1 = 1.531\,10^5, m_2 = 8165\right]$:

Derive Expressions for the Optimum Spring and Damping Constant of the Tuned Mass Damper

Mass ratio

> $\mu := \frac{m_2}{m_1}$:

Natural frequency of tuned mass damper

> $\omega_2 := \sqrt{\frac{k_{2\_calc}}{m_2}}$:

Natural frequency of main system

> $\omega_1 := \sqrt{\frac{k_1}{m_1}}$:

Hence the natural frequency in rad s^{-1} is

> eval(ω_1, params)

$$13.98492872 \tag{2.1}$$

Ratio of natural frequencies

> $\alpha := \frac{\omega_2}{\omega_1}$:

Optimum ratio of natural frequencies

> $\alpha_{opt} := \frac{1}{1+\mu}$:

Hence the optimum spring constant of the tuned mass-spring-damper

> k_{2_calc} := solve($\alpha = \alpha_{opt}$, k_{2_calc})

$$k_{2\_calc} := \frac{m_1 k_1 m_2}{(m_1 + m_2)^2} \tag{2.2}$$

Damping Ratio

> $z := \frac{b_{2\_calc}}{2 m_2 \omega_2}$:

Optimum damping ratio

> $z_{opt} := \sqrt{\frac{3\mu}{8(1+\mu)^3}}$:

Hence the optimum damping constant of the tuned mass-spring-damper

> b_{2_calc} := solve($z = z_{opt}$, b_{2_calc})

$$b_{2\_calc} := \frac{1}{2}\sqrt{6}\sqrt{\frac{m_2}{m_1\left(1+\frac{m_2}{m_1}\right)^3}}\, m_2 \sqrt{\frac{m_1 k_1}{(m_1+m_2)^2}} \tag{2.3}$$

> k_{2_calc} := eval(k_{2_calc}, params);

$$k_{2\_calc} := 1.458730861\ 10^6 \tag{2.4}$$

> b_{2_calc} := evalf(eval(b_{2_calc}, params));

$$b_{2\_calc} := 26869.77094 \tag{2.5}$$

Full list of parameters

> $params_{tuned} := \left[m_1 = 1.764\,10^5, k_1 = 3.45\,10^7, b_1 = 1.531\,10^5, m_2 = 8165, k_2 = k_{2\_calc}, b_2 = b_{2\_calc}\right]$:

> $params_{nottuned} := \left[m_1 = 1.764\,10^5, k_1 = 3.45\,10^7, b_1 = 1.531\,10^5, m_2 = 0, k_2 = k_{2\_calc}, b_2 = b_{2\_calc}\right]$:

Equations of Motion for the Entire System

Equation of motion for the whole system

> de := $m_2 \left(\frac{d^2}{dt^2} x_2(t)\right) = -k_2 \left(x_2(t) - x_1(t)\right) - b_2 \left(\frac{d}{dt} x_2(t) - \frac{d}{dt} x_1(t)\right),$

$m_1 \left(\frac{d^2}{dt^2} x_1(t)\right) = -k_1 x_1(t) - b_1 \left(\frac{d}{dt} x_1(t)\right) - k_2 \left(x_1(t) - x_2(t)\right) - b_2 \left(\frac{d}{dt} x_1(t) - \frac{d}{dt} x_2(t)\right) + F(t)$:

ic := $x_1(0) = 0, D(x_1)(0) = 0, x_2(0) = 0, D(x_2)(0) = 0$:

> sys := DiffEquation([de], [F(t)], [$x_1(t)$]) :

Frequency Response

Response with Tuned Mass Damper

> p1 := MagnitudePlot(sys, range = 5 ..30, parameters = $params_{tuned}$, color = blue, legend = "Tuned") :

Response with no Tuned Mass Damper

> p2 := MagnitudePlot(sys, range = 5 ..30, parameters = $params_{nottuned}$, color = red, axesfont = [Calibri], labelfont = [Calibri], legend = "Not Tuned") :

> display(p1, p2)



Dynamic Response

Assuming that the main system is perturbed at its natural frequency

> f := eval($\left(\omega_1\right.$, params)

$$f := 13.98492872 \tag{5.1}$$

> p1 := ResponsePlot(sys, 7500 sin(f·t), parameters = $params_{tuned}$, color = blue, numpoints = 2^{10}, legend = "Tuned") :

> p2 := ResponsePlot(sys, 7500 sin(f t), parameters = $params_{nottuned}$, color = red, numpoints = 2^{10}, legend = "Not Tuned") :

> display(p1, p2, axesfont = [Calibri], labelfont = [Calibri], axesfont = [Calibri], *gridlines*)



Power Spectrum

> tunedResponseData := plottools:-getdata(p1) [3] :

> notTunedResponseData := plottools:-getdata(p2) [3] :

> $\text{samplingRate} := \dfrac{1}{\text{tunedResponseData}[2, 1] - \text{tunedResponseData}[1, 1]}$:

> psTuned := PowerSpectrum(FFT(tunedResponseData[.., 2])) :

> psNotTuned := PowerSpectrum(FFT(notTunedResponseData[.., 2])) :

> $\text{psPlot1} := \text{pointplot}\left(\left[\text{seq}\left(\left[\dfrac{i \cdot \text{samplingRate}}{2^{10}} \cdot 2 \cdot \text{Pi}, \text{psTuned}[i]\right], i = 1 .. \dfrac{2^{10}}{2}\right)\right], \text{connect} = \text{true}, \text{legend} = \text{"Tuned"}, \text{gridlines}\right)$:

$\text{psPlot2} := \text{pointplot}\left(\left[\text{seq}\left(\left[\dfrac{i \cdot \text{samplingRate}}{2^{10}} \cdot 2 \cdot \text{Pi}, \text{psNotTuned}[i]\right], i = 1 .. \dfrac{2^{10}}{2}\right)\right], \text{connect} = \text{true}, \text{color} = \text{red}, \text{legend} = \text{"Not Tuned"}\right)$:

> display(psPlot1, psPlot2, axis[1] = [mode = log], axesfont = [Calibri], labels = ["Frequency [rad/s]", "Power"], labeldirections = [horizontal, vertical], labelfont = [Calibri])


0.0015
0.0010
0.0005
0
Power
1
5
10
50
100
Frequency [rad/s]
Tuned
Not Tuned

>